

Linguagem Csmall

Gramática (definições sintáticas):

Programa \rightarrow INT MAIN LBRACKET RBRACKET LBRACE Decl_Comando RBRACE
Decl_Comando \rightarrow Declaracao Decl_Comando | Comando Decl_Comando | ϵ
Declaracao \rightarrow Tipo ID Decl2
Decl2 \rightarrow COMMA ID Decl2 | PCOMMA | ATTR Expressao Decl2
Tipo \rightarrow INT | FLOAT
Comando \rightarrow Bloco | Atribuicao | ComandoSe | ComandoEnquanto | ComandoRead |
ComandoPrint | ComandoFor
Bloco \rightarrow LBRACE Decl_Comando RBRACE
Atribuicao \rightarrow ID ATTR Expressao PCOMMA
ComandoSe \rightarrow IF LBRACKET Expressao RBRACKET Comando ComandoSenao
ComandoSenao \rightarrow ELSE Comando | ϵ
ComandoEnquanto \rightarrow WHILE LBRACKET Expressao RBRACKET Comando
ComandoRead \rightarrow READ ID PCOMMA
ComandoPrint \rightarrow PRINT LBRACKET Expressao RBRACKET PCOMMA
ComandoFor \rightarrow FOR LBRACKET AtribuicaoFor PCOMMA Expressao PCOMMA
AtribuicaoFor RBRACKET Comando
AtribuicaoFor \rightarrow ID ATTR Expressao
Expressao \rightarrow Conjuncao ExpressaoOpc
ExpressaoOpc \rightarrow OR Conjuncao ExpressaoOpc | ϵ
Conjuncao \rightarrow Igualdade ConjuncaoOpc
ConjuncaoOpc \rightarrow AND Igualdade ConjuncaoOpc | ϵ
Igualdade \rightarrow Relacao IgualdadeOpc
IgualdadeOpc \rightarrow OpIgual Relacao IgualdadeOpc | ϵ
OpIgual \rightarrow EQ | NE
Relacao \rightarrow Adicao RelacaoOpc
RelacaoOpc \rightarrow OpRel Adicao RelacaoOpc | ϵ
OpRel \rightarrow LT | LE | GT | GE
Adicao \rightarrow Termo AdicaoOpc
AdicaoOpc \rightarrow OpAdicao Termo AdicaoOpc | ϵ
OpAdicao \rightarrow PLUS | MINUS
Termo \rightarrow Fator TermoOpc
TermoOpc \rightarrow OpMult Fator TermoOpc | ϵ
OpMult \rightarrow MULT | DIV
Fator \rightarrow ID | INTEGER_CONST | FLOAT_CONST | LBRACKET Expressao
RBRACKET

Definições léxicas:

ID \rightarrow [A-Za-z] ([A-Za-z] | [0-9] | $_$)*
INTEGER_CONST \rightarrow [0-9] ([0-9])*
FLOAT_CONST \rightarrow [0-9] ([0-9])*.[0-9] ([0-9])*

